TB 13-18 Flow between plates

3-30-18

N. T. Gladd

Initialization: Be sure the file *NTGUtilityFunctions.m* is in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

This is the 11th in a series of notebooks in which I work through material and exercises in the magisterial new book *Modern Classical Physics* by Kip S. Thorne and Roger D. Blandford. If you are a physicist of any ilk, BUY THIS BOOK. You will learn from a close reading and from solving the exercises.

Exercise 13.18 *Problem: Steady Flow between Two Plates*

A viscous fluid flows steadily (no time dependence) in the z direction, with the flow confined between two plates that are parallel to the x-z plane and are separated by a distance $2a$. Show that the flow's velocity field is

$$v_z = -\frac{dP}{dz}\frac{a^2}{2\eta}\left[1-\left(\frac{y}{a}\right)^2\right],$$

and the mass flow rate (the discharge) per unit width of the plates is

$$\frac{dm}{dt\,dx} = -\frac{dP}{dz}\frac{2\rho a^3}{3\eta}.$$

Here dP/dz (which is negative) is the pressure gradient along the direction of flow. (In Sec. 19.4 we return to this problem, augmented by a magnetic field and electric current, and discover great added richness.)



The solution of this problem is analogous to the pipe flow example in section 13.7.6 of the text (see below).

```
w1[1] =
 1/ρ Laplacian[vz[y], {x, y, z}, "Cartesian"] == 1/η Grad[P[z], {x, y, z}, "Cylindrical"][[3]]
```

$$\frac{vz''[y]}{\rho} == \frac{P'[z]}{\eta}$$

```
w1[2] = DSolve[{w1[1], vz[-a] == 0, vz[a] == 0}, vz[y], y][[1, 1]] // Simplify
```

$$vz[y] \to \frac{\left(-a^2 + y^2\right)\, \rho\, P'[z]}{2\,\eta}$$

which is 13.82a

The mass per time flowing through a particular z level is

$$\frac{dm}{dt} = \int_A dA\, \rho\, vz \qquad (1)$$

Thus

```
w1[4] = dmdt == Integrate[((-a^2 + y^2) ρ P'[z])/(2 η), {y, -a, a}, {x, 0, Lx}]
```

$$dmdt == -\frac{2 a^3 Lx \rho P'[z]}{3 \eta}$$

and the flow per length Lx of the plate is

```
w1[5] = MapEqn[(# / Lx) &, w1[4]]
```

$$\frac{dmdt}{Lx} == -\frac{2 a^3 \rho P'[z]}{3 \eta}$$

which is 13.82 b).

Pipe flow basics

I work through the example of flow in a pipe in Section 13.7.6

Flow in a pipe





Consider Navier-Stokes equation with gravity assumed negligible

$$\frac{\partial \boldsymbol{v}}{\partial t} + (\boldsymbol{v} \cdot \nabla) \cdot \boldsymbol{v} = -\frac{\nabla P}{\rho} + \frac{1}{\rho} \frac{\partial \sigma_{\mathrm{ik}}}{\partial x_k} \tag{2}$$

For incompressible flow, this can be written

$$\frac{\partial \boldsymbol{v}}{\partial t} + (\boldsymbol{v} \cdot \nabla) \cdot \boldsymbol{v} = -\frac{\nabla P}{\rho} + \alpha \nabla^2 \boldsymbol{v} \tag{3}$$

where I use $\alpha = \eta/\rho$ to represent the kinetic viscosity. The Mathematica font for “nu” looks too much like “vee”.

Consider time stationary flow through a pipe — implies $\partial \mathbf{v}/\partial t = 0$.

$$(\boldsymbol{v} \cdot \nabla) \cdot \boldsymbol{v} = -\frac{\nabla P}{\rho} + \alpha \nabla^2 \boldsymbol{v} \tag{4}$$

Assume flow in laminar in the pipe $\mathbf{v} = v_z(\mathrm{R})\ \mathbb{1}_z$ — implies $(v_z(\mathrm{R}) \frac{\partial}{\partial z}) v_z(\mathrm{R}) = 0$.

Assume the pressure depends only z

$$\frac{\nabla_z P}{\rho} = \alpha \nabla^2 v_z \tag{5}$$

```
w2[1] = 1/ρ Laplacian[vz[R], {R, θ, z}, "Cylindrical"] ==
   1/η Grad[P[z], {R, θ, z}, "Cylindrical"][[3]]
```

$$\frac{\frac{vz'[R]}{R} + vz''[R]}{\rho} == \frac{P'[z]}{\eta}$$

where

```
w2[2] = DSolve[w2[1], vz[R], R][[1, 1]]
```

$$vz[R] \to C[2] + C[1]\,Log[R] + \frac{R^2\,\rho\,P'[z]}{4\,\eta}$$

Boundary condition is vz[a] = 0.

```
w2[3] = w2[2] /. R → a /. vz[a] → 0 // RE
```

$$0 == C[2] + C[1]\,Log[a] + \frac{a^2\,\rho\,P'[z]}{4\,\eta}$$

```
w2[4] = Sol[w2[3], C[2]]
```

$$C[2] \to \frac{-4\,\eta\,C[1]\,Log[a] - a^2\,\rho\,P'[z]}{4\,\eta}$$

```
w2[5] = w2[2] /. w2[4] // Expand
```

$$vz[R] \to -C[1]\,Log[a] + C[1]\,Log[R] - \frac{a^2\,\rho\,P'[z]}{4\,\eta} + \frac{R^2\,\rho\,P'[z]}{4\,\eta}$$

Symmetry condition is vz’[0] = 0

```
w2[6] = (D[#, R]) & /@ w2[5]
```

$$vz'[R] \to \frac{C[1]}{R} + \frac{R\,\rho\,P'[z]}{2\,\eta}$$

For For v’[R] → 0 and R → 0, must have C[1] = 0

```
w2[7] = w2[5] /. C[1] → 0 // Simplify
```

$$vz[R] \to \frac{\left(-a^2 + R^2\right)\,\rho\,P'[z]}{4\,\eta}$$

Calculate the flow rate

```
w2[8] = 𝓕 == Integrate[((-a^2 + R^2) ρ P′[z])/(4 η) 2 π R, {R, 0, a}]
```

$$\mathcal{F} == -\frac{a^4 \pi \rho P'[z]}{8 \eta}$$

which is TB 13.81.

Visualizations

```
Module[{a = 1, η = 1, dPdz = -1, w = 0.1,
  Lx = 2, Lz = 3, P = {0, 0, 1.5}, 𝕚x = {1, 0, 0}, 𝕚y = {0, 1, 0},
  𝕚z = {0, 0, 1}, axes, plate, fluid, flowVectors, Axis, VzFlow},
 Axis[OO_, P_, lab_, mult_, color_] :=
  {Directive[color, Thick], Line[{OO, P}], Text[Stl[lab], mult P]};
 axes = With[{mult = 1, color = LightGray},
   Axis[P, #[[1]], #[[2]], mult, color] & /@
    {{P + 0.25 𝕚x, "x"}, {P + 0.25 𝕚y, "y"}, {P + 0.25 𝕚z, "z"}}];

 VzFlow[y_, a_, η_, dPdz_] := 1/2 (-a^2 + y^2) η dPdz;
 plate[1] = {Gray, EdgeForm[Thick], Cuboid[{0, -a, 0}, {Lx, -a + w, Lz}]};
 plate[2] = {Gray, Cuboid[{0, a, 0}, {Lx, a + w, Lz}]};
 fluid = {Blue, Opacity[0.5], Cuboid[{0, -a, 0}, {Lx, a, Lz}]};
 flowVectors = {Arrowheads[Small], White,
   Table[Arrow[{{x, y, 0}, {x, y, VzFlow[y, a, η, dPdz]}}],
    {y, -a, a, a / 4}, {x, 0, Lx, Lx / 2}]};
 Graphics3D[{axes, plate[1], plate[2], fluid, flowVectors},
  Axes → None, AxesLabel → {"x", "y", "z"},
  PlotRange → {{-0.25, Lx + 0.25}, {-1.25, 1.25}, {-0.25, Lz + 0.25}},
  PlotLabel → Stl["Flow between plates"], Boxed → False]]
```

Flow between plates




```
CoordinateChartData[{"Polar", 2}, "CoordinateRangeAssumptions", {R, θ}]
```

$R > 0 \,\&\&\, -\pi < \theta \leq \pi$

```
Module[{R = 1, L = 3, α = 4, dPdz = -1, O, P, pipe, flowVectors, CyltoCar, VzFlow},
 O = {0, 0, 0}; P = {0, 0, L};
 pipe = {{Blue, EdgeForm[], Opacity[0.5], Cylinder[{O, P}, R]},
   {Gray, EdgeForm[], Opacity[0.5], Cylinder[{O, P}, 1.1 R]},};
 CyltoCar[R_, θ_, z_] := CoordinateTransform[
   "Cylindrical" → "Cartesian", { R, θ, z}];

 VzFlow[R_, α_, dPdz_] := 1/4 (-1 + R^2) α dPdz;
 flowVectors = {Arrowheads[Small], White,
   Table[Arrow[{CyltoCar[R, θ, 0], CyltoCar[R, θ, VzFlow[R, α, dPdz]]}],
    {θ, -π + π/8, π, π/8}, {R, 0.01, 1.0, 0.2}]};
 Graphics3D[{pipe, flowVectors}, PlotLabel → Stl["Flow in a pipe"],
  Boxed → False, ImageSize → 300]]
```

Flow in a pipe



Cylindrical geometry

```
Module[{O = {0, 0, 0}, 𝟙x = {1, 0, 0}, 𝟙y = {0, 1, 0}, 𝟙z = {0, 0, 1},
  axes, range, P, Q, ϕArc, refLines, CyltoCar, Axis, ArcArrow3D, G},
 CyltoCar[R_, θ_, z_] := CoordinateTransform[
   "Cylindrical" → "Cartesian", { R, θ, z}];
 Axis[OO_, P_, lab_, mult_, color_] :=
  {Directive[color, Thick], Line[{OO, P}], Text[Stl[lab], mult P]};
 ArcArrow3D[R_, θS_, θF_, z_] :=
  {Arrow@Table[CyltoCar[R, θ, z], {θ, θS, θF, π/64}]};

 axes = With[{mult = 1.2, color = LightGray},
   Axis[O, #[[1]], #[[2]], mult, color] & /@ {{𝟙x, "x"}, {𝟙y, "y"}, {𝟙z, "z"}}];
 range = 1.2 {{-0.1, 1}, {-0.1, 1}, {-0.1, 1}};
 {P, Q} =
  With[{R = 0.8, θ = 3 π/8, z = 0.8}, {CyltoCar[R, θ, z], CyltoCar[R, θ, 0]}];
 refLines = {Black, Line[{P, Q}], Line[{O, Q}], Point[P], Text[Stl["P"], 1.1 P],
   Text[Stl["z"], (P + Q)/2 + {0, 0.1, 0}], Text[Stl["R"], Q/2 + {0, 0, 0.1}]};
 ϕArc = With[{R = 0.4, θ = 3 π/8, z = 0}, {Black, Arrowheads[0.02],
    ArcArrow3D[R, 0, θ, 0], Text["θ", CyltoCar[0.7 R, θ/2, z]]}];
 G[1] = Graphics3D[{axes, refLines, ϕArc}, BoxRatios → Automatic,
   Boxed → False, Axes → None, PlotRange → range,
   ViewPoint → {2, 2, 1}, PlotLabel → Stl["cylindrical geometry"] ];
 G[2] = With[{R = 0.8}, ParametricPlot3D[{{R Cos[θ], R Sin[θ], z}}, {θ, 0, π/2},
    {z, 0, 1}, Mesh → False, ColorFunction → "TemperatureMap", PlotStyle → Opacity[.25],
    Axes → None, Boxed → False, ViewPoint → {2, 2, 1}, PlotRange → range]];
 Show[
  G[
   1],
  G[
   2]]]
```


cylindrical geometry
z
P
z
R
θ
x
y